QT       += core gui httpserver

greaterThan(QT_MAJOR_VERSION, 4): QT += widgets

CONFIG += c++17

# You can make your code fail to compile if it uses deprecated APIs.
# In order to do so, uncomment the following line.
#DEFINES += QT_DISABLE_DEPRECATED_BEFORE=0x060000    # disables all the APIs deprecated before Qt 6.0.0

SOURCES += \
    formatter.cpp \
    idreader.cpp \
    logging.cpp \
    main.cpp \
    mwlserver.cpp \
    settings.cpp

HEADERS += \
    formatter.h \
    idreader.h \
    logging.h \
    mwlserver.h \
    settings.h

DCMTK_PATH = C:/Users/zhumangen/workspace/3rd/dcmtk-3.6.8
INCLUDEPATH *= $${DCMTK_PATH}/dcmwlm/include
INCLUDEPATH *= $${DCMTK_PATH}/build/Desktop_Qt_6_7_2_MinGW_64_bit-Release2/config/include
INCLUDEPATH *= $${DCMTK_PATH}/ofstd/include
INCLUDEPATH *= $${DCMTK_PATH}/dcmdata/include
INCLUDEPATH *= $${DCMTK_PATH}/dcmnet/include
INCLUDEPATH *= $${DCMTK_PATH}/oflog/include

DEPENDENT_LIBRARIES    *= dcmwlm \
                          oflog \
                          dcmnet \
                          dcmdata \
                          ofstd \
                          oficonv \
                          ws2_32 \
                          netapi32 \
                          iphlpapi

CONFIG(debug, debug|release) {
    LIBS *= -L$$quote($${DCMTK_PATH}/build/Desktop_Qt_6_7_2_MinGW_64_bit-Debug2/lib)
} else {
    LIBS *= -L$$quote($${DCMTK_PATH}/build/Desktop_Qt_6_7_2_MinGW_64_bit-Release2/lib)
}

for(library, DEPENDENT_LIBRARIES) {
    LIBS               *= -l$$quote($$library)
}

# Default rules for deployment.
qnx: target.path = /tmp/$${TARGET}/bin
else: unix:!android: target.path = /opt/$${TARGET}/bin
!isEmpty(target.path): INSTALLS += target

RC_FILE = logo/main.rc
